SPIRAL CURVE SOLUTIONS

FOR THE HP-41CV/CX

```
         9+00.000       [illegible]
         9+25.000       107.88
         9+50.000       107.50
         9+75.000       107.13
        10+00.000       106.75
        10+25.000       106.38

      10+50.000 BVC   106.00
       200 FT V.C.
      11+50.000 PVI   104.50
                POVC  105.88
      11+04.545 G=0%  105.59
```

prompt: STA?

keystrokes:

2 **5** **1**

output:

```
        10+75.000       105.71
        11+00.000       105.59
        11+25.000       105.65
```



Software by D'Zign
P. O. BOX 1370 • PACIFICA, CA 94044

NOTICE

SPIRAL CURVE SOLUTIONS

for the HP-41CV/CX

ISBN 0-944889-00-X

published by D'Zign Land Survey & Development
Pacifica, California 1987

Introduction

This group of **SPIRAL CURVE** programs is designed to calculate the data needed for field layout of a spiral curve. The type of spiral used is the BARNETT SPIRAL, (also known as the Talbot spiral). It uses arc definition, and is the type most frequently encountered.

This is the form of spiral used on interstate highways, and adopted by most states which use spiral curves in their alignment design. It is also the form of spiral used for railroad alignment by Rapid Transit Agencies such as Washington D.C.'s METRO, the Bay Area Rapid Transit District (BART), and the new systems being constructed in Seattle and Los Angeles.

The nomenclature used for the spiral system, and the alignment data from a typical set of plans are shown below.



SPIRAL SYSTEM

ALIGNMENT DATA- ℄

	POINT	STATION	COORDINATES N	COORDINATES E	CURVE DATA	DESIGN SPEED
M2-LINE	POT	130+99.549	38,196.113	30,742.993	P.I. STA = 140+36.445	
	PI-4		39,026.499	30,394.100	Δ = 22° 06' 55" T_{S1} = 418.175' T_{S2} = 418.175'	
	T S	136+18.270	38,638.388	30,549.792	L_{S1} = 265.000' X_{S1} = 264.782	V = 50 M.P.H. E = 5.250" U = 2.710"
	SC	138+83.270	38,887.117	30,458.646	θ_{S1} = 5° 11' 59.2", Y_{S1} = 8.012'	
		CC-CURVE 4	39,305.651	31,857.310	Δc = 11° 42' 56.5" L_c = 298.537' R_c = 1460.000'	
	CS	141+81.807	39,179.855	30,402.800	L_{S2} = 265.000' X_{S2} = 264.782'	
	ST	144+46.807	39,444.670	30,395.975	θ_{S2} = 5° 11' 59.2" Y_{S2} = 8.012'	
	PI-10	151+27.015	40,124.871	30,399.025		

Deflection & chord

In the first routine using this program the **CHORD** and **DEFLECTION ANGLE** to any station on the spiral is calculated. The radial angle at the station is also calculated.

To illustrate this routine, the spiral data on the previous page will be used. The entrance spiral is shown below, and the curve data for the spiral portions is the same for both the entrance and exit spirals. Stationing at the **CS** and **ST** will be output by the program after calculations for the entrance spiral have been completed.





The keys used for this routine are shown in the sketch to the left.

The required input is prompted by the program as you proceed, and followed by stroking the button, except after input of a new station for solution.

The program may be used without a printer, but for ease of illustration, the printer output will be shown.

Step-by-step instructions begin on the next page.

User Instructions

Program memory should contain the program "**SP**" and the utility programs "**DMS**", "**STA**" and "**AZ**" before you begin. "**AZ**" may be omitted if the calculator contains the HP Surveying Pac. Size the calculator at a minimum of 050.

Initialize the program by stroking XEQ ALPHA S P ALPHA

The program will begin the prompts to determine which type of solution is wanted.

1 **COORD-O/S?**

Answer this prompt no with N R/S

2 **TAN O/S?**

Answer this prompt no, stroke N R/S

3 **PI STATION?**

Input the main P.I. station. If it is not shown on the curve data provided it can be calculated by inputting the TS station and adding the T_S distance to it.

4 **DELTA?**

Input the system delta. If the curve is to the left, first stroke CHS, then R/S

5 **R?**

Input the radius of the circular curve, then stroke R/S

6 **L?**

Input the spiral length R/S

Output will be a display of the length of spiral curve, the spiral angle, and the radius. If a printer is not attached, continue stroking R/S. Output continues with the P.I. station, the central angle, and the TS and SC stations, followed by the next prompt:

7 **STA?**

Input the station for which the deflection and chord solution is required E

Output will be the chord, deflection angle and radial angle. Continue stroking R/S after each portion of the output if you are not using a printer, until the prompt STA? appears.

8 **STA?**

Repeat step 7 until all of the required stations have been calculated for the entrance spiral. It is normal to also calculate the SC station last. When ready to calculate the exit spiral, stroke D

Output will be the stations of the CS and ST, followed by the prompt,

9 **STA?**

Input the exit spiral stations for solution in the same manner as for the entrance spiral, by stroking E

Repeat step 7 until all of the required stations have been calculated. The exit spiral can be calculated in either direction, but the deflection angles and chords are from the ST, sighting toward the P.I.

As an example of the keystrokes and output used with this routine, we will use the information from the spiral system shown on page 3. We will calculate the 100' stations through both the entrance and exit spirals.

Even if this is not the form of solution which you would normally use (spraying by coordinates seems to be most widely used), we recommend that you try this example, both for practice, and to insure that your program is working correctly.

Remember to continue stroking R/S through the output, until the next prompt appears, if you are using the calculator without a printer attached.

keystrokes:

XEQ ALPHA S P ALPHA

prompt: **COORD-O/S?**

keystrokes:

N R/S

prompt: **TAN O/S?**

keystrokes:

N R/S

prompt: **PI STATION?**

keystrokes:

1 4 0 3 6 . 4 4 5 R/S

prompt: **DELTA?**

keystrokes:

2 2 . 0 6 5 5 R/S

prompt: **R?**

keystrokes:

1 4 6 0 R/S

prompt: **L?**

keystrokes:

2 6 5 R/S

output:

L = 265.0000
S∡=5°11'59.2"
R = 1,460.0000

PI=140+36.445
CENTRAL ∡ = 22°06'55.0"

TS=136+18.270
SC=138+83.270

At this point we begin to calculate the even stations along the entrance spiral

prompt: **STA?**

keystrokes:

1 3 7 0 0 E

output:

137+00.000
CD = 81.730
DEFLECTION ∡=0°09'53.5"
RADIAL ∡ =90°19'47.0"

prompt: **STA?**

keystrokes:

1 3 8 0 0 E

output:

138+00.000
CD = 181.715
DEFLECTION ∡=0°48'54.4"
RADIAL ∡ =91°37'49.0"

prompt: **STA?**

keystrokes:

1 3 8 8 3 . 2 7 E

output:

138+83.270
CD = 264.903
DEFLECTION ∡=1°43'59.3"
RADIAL ∡ =93°27'59.9"

With the calculations for the entrance spiral now completed, we can move to the exit spiral

keystroke:

D

output:

ST=144+46.807
CS=141+81.807

We will continue to go in the upstation direction for the exit spiral calculations

prompt: **STA?**

keystrokes:

1 4 1 8 1 · 8 0 7 E

output:

141+81.807
CD = 264.903
DEFLECTION ∠=1°43'59.3"
RADIAL ∠ =93°27'60.0"

keystrokes:

1 4 2 0 0 E

output:

142+00.000
CD = 246.739
DEFLECTION ∠=1°30'12.1"
RADIAL ∠ =93°00'25.1"

keystrokes:

1 4 3 0 0 E

output:

143+00.000
CD = 146.802
DEFLECTION ∠=0°31'55.0"
RADIAL ∠ =91°03'50.0"

keystrokes:

1 4 4 0 0 E

output:

144+00.000
CD = 46.807
DEFLECTION ∠=0°03'14.7"
RADIAL ∠ =90°06'29.3"

To calculate a new curve, XEQ ALPHA S P ALPHA, and begin input of the data for the new curve.

Deflections and chords for the circular portion may be calculated as a normal curve, and may be staked with the instrument at the SC. Sight the CS with the total deflection angle set in the instrument.

Alignment & Offsets

This routine outputs the coordinates along a spiral curve alignment at any station, including the circular portion. It not only provides direct output of the centerline coordinates, but the coordinates of any offsets to the curves may be calculated at the same time.

In addition to the information about the spiral system which was input in the last routine, you will need to know the coordinates of the **TS** and the **ST** stations, and the bearings of the entrance and exit tangents. If these are not given on the set of plans that you are working from, they can be easily calculated prior to beginning this routine.

The keys used in this routine are shown in the keyboard assignment sketch below.





Begin with the calculator sized at 050, and program memory containing the programs "**SP**", "**STA**","**DMS**" and "**AZ**" (the latter may be omitted if the calculator contains a ROM module for the HP Surveying Pac). After being initialized the program will begin with the prompts for type of solution wanted.

Initialize the program by stroking XEQ ALPHA S P ALPHA

1 **COORD-O/S?**
Answer this prompt yes, Y R/S

2 **BRG=?**
Input the entrance tangent bearing R/S

3 **QD=?**
Input the quadrant code for the bearing **toward** the P.I. of the system R/S

4 **TS N↑E**
Input the north coordinate of the TS ENTER↑

Input the east coordinate of the TS R/S

5 **INVERSE?**
Answer no, with N R/S

6 **PI STATION?**
Input the main P.I. station. R/S

7 **DELTA?**
Input the system delta. If curve left, first stroke CHS, then R/S

8 **R?**
Input the radius for the circular curve R/S

9 **L?**
Input the spiral length R/S

Output will be a display of the length of spiral curve, the spiral angle and the radius. If a printer is not attached, continue stroking R/S. Output continues with the P.I. station, the central angle, and the TS and SC stations, followed by the next prompt

10 **STA?**
Input the station for which coordinates are required
E

Output will be the station and its coordinates. Continue stroking R/S each time, if not using a printer, until the prompt **O/S DIST?** appears.

11 **O/S DIST?**
Any desired offsets to this station may be calculated at this time. Input the required offset distance (CHS if the offset is to the left)
R/S

Output will be the offset and its coordinates. An offset to the left will be shown as a negative offset.

12 **O/S DIST?**
Repeat step 11 until all of the required offsets for the station have been calculated, or return to step 10 with input of a new station.

When all of the required stations and offsets have been calculated for the entrance spiral, we can go to the circular portion, as follows:

13 **O/S DIST?**
Calculate the SC station last. When you are ready to calculate the circular portion, keystroke
J

14 **O/S DIST?**
Input the circular radius distance. If the curve is to the left, CHS
R/S

Output will be the coordinates of the radius point of the circular curve. **The circular portion has a slightly different input format than the spiral portion.** The station will be input each time, for each offset. For the centerline station coordinates, the offset is input as 0.

15 **STA?**
Input the next station
R/S

16 **O/S DIST?**

Input 0 for the centerline coordinates, or the offset distance. If the offset is to the left of centerline, stroke CHS before stroking

Output will be the station and its coordinates (or the offset and its coordinates).

17 **STA?**

Repeat steps 15 and 16 until all of the stations and offsets have been calculated through the circular portion. Go to the exit spiral by keystroking D

18 **BRG=?**

Input the bearing of the **exit** tangent R/S

19 **QD=?**

Input the quadrant code for the exit tangent in the direction **toward** the P.I. R/S

20 **ST N↑E**

Input the north coordinate of the ST ENTER↑

Input the east coordinate of the ST R/S

Output will be the ST and CS stations.

21 **STA?**

Calculate the stations through the exit spiral, beginning with the CS station, by repeating steps 10 and 11 until all of the required stations and offsets have been calculated.

The coordinates for the curve points are usually given on the plans for the project, and it is a good idea to calculate all of these points when you use this routine, as a check. Note that the example data on page 3 even gives the coordinates for the radius point of the circular curve (cc-curve 4).

Keystroke Examples

As an example of the keystrokes used with this routine, and using the information on page 3, in the example spiral, we will calculate the coordinates at even stations. In addition, to use the offset option, we will calculate the coordinates for an offset at 20 feet left and right at one of the stations in the entrance, circular and exit portions of the system.

keystrokes:

XEQ ALPHA S P ALPHA

prompt: **COORD-O/S?**

keystrokes:

Y R/S

prompt: **BRG=?**

keystrokes:

2 1 . 5 1 3 R/S

prompt: **QD=?**

keystrokes:

4 R/S

prompt: **TS N↑E**

keystrokes:

3 8 6 3 8 .
3 8 8 ENTER↑
3 0 5 4 9 .
7 9 2 R/S

prompt: **INVERSE?**

keystrokes:

N R/S

prompt: **PI STATION?**

keystrokes:

1 4 0 3 6 .
4 4 5 R/S

prompt: **DELTA?**

keystrokes:

2 2 . 0 6 5 5
R/S

prompt: **R?**

keystrokes:

1 4 6 0 R/S

prompt: **L?**

keystrokes:

2 6 5 R/S

output:

```
L = 265.0000
S∡=5°11'59.2"
R = 1,460.0000

PI=140+36.445
CENTRAL ∡ = 22°06'55.0"

TS=136+18.270
SC=138+83.270
```

prompt: **STA?**

keystrokes:

1 3 7 0 0 E

output:

```
137+00.000
N = 38,714.3292
E = 30,519.5814
```

prompt: **O/S DIST?**

we will use this station as an example for the offsets.

keystrokes:

[2] [0] [CHS] [R/S]

output:

O/S = -20.0000
N = 38,707.0434
E = 30,500.9556

prompt: **O/S DIST?**

keystrokes:

[2] [0] [R/S]

output:

O/S = 20.0000
N = 38,721.6149
E = 30,538.2071

prompt: **O/S DIST?**

keystrokes:

[1] [3] [8] [0] [0] [E]

output:

138+00.000
N = 38,807.9846
E = 30,484.5431

prompt: **O/S DIST?**

keystrokes:

[1] [3] [8] [8] [3] [.] [2] [7] [E]

output:

138+83.270
N = 38,887.1168
E = 30,458.6462

prompt: **O/S DIST?**

After calculating any needed offsets at the **SC**, move to the circular portion of the system



keystroke:

[J]

output:

CIRCULAR:

prompt: **O/S DIST?**

keystrokes:

[1] [4] [6] [0] [R/S]

output:

RADIUS POINT:
N = 39,305.6513
E = 31,857.3702

Keystroke Examples

prompt: **STA?**

keystrokes:

1 3 9 0 0 R/S

output:

139+00.000

prompt: **O/S** =

keystrokes:

0 R/S

output:

N = 38,903.1717
E = 30,453.9422

prompt: **STA?**

keystrokes:

1 3 9 0 0 R/S

output:

139+00.000

prompt: **O/S** =

keystrokes:

2 0 CHS R/S

output:

O/S = -20.0000
N = 38,897.6583
E = 30,434.7171

prompt: **STA?**

keystrokes:

1 3 9 0 0 R/S

output:

139+00.000

prompt: **O/S** =

keystrokes:

2 0 R/S

output:

O/S = 20.0000
N = 38,908.6852
E = 30,473.1672

prompt: **STA?**

keystrokes:

1 4 0 0 0 R/S

output:

140+00.000

prompt: **O/S** =

keystrokes:

0 R/S

output:

N = 39,000.1655
E = 30,429.6873

prompt: **STA?**

keystrokes:

1 4 1 0 0 R/S

output:

141+00.000

prompt: **O/S** =

keystrokes:

0 R/S

output:

N = 39,098.5919
E = 30,412.1275

prompt: **STA?**

With the circular portion completed, go to the exit spiral

keystroke:

D

prompt: **BRG=?**

keystrokes:

· 1 5 2 5 R/S

prompt: **QD=?**

keystrokes:

[4] [R/S]

prompt: **ST N↑E**

keystrokes:

[3] [9] [4] [4] [4] [.]
[6] [7] [ENTER↑]
[3] [0] [3] [9] [5] [.]
[9] [7] [5] [R/S]

output:

ST=144+46.807
CS=141+81.807

prompt: **STA?**

keystrokes:

[1] [4] [1] [8] [1] [.]
[8] [0] [7] [E]

output:

141+81.807
N = 39,179.8546
E = 30,402.7994

prompt: **O/S** =

Since all of the offsets through the exit spiral work the same at any station, we can use the **CS** as the example station for the offset calculations

keystrokes:

[2] [0] [CHS] [R/S]

output:

O/S = -20.0000
N = 39,178.1314
E = 30,382.8737

prompt: **O/S**

keystrokes:

[2] [0] [R/S]

output:

O/S = 20.0000
N = 39,181.5779
E = 30,422.7250

prompt: **O/S DIST?**

keystrokes:

[1] [4] [2] [0] [0] [E]

output:

142+00.000
N = 39,197.9891
E = 30,401.3422

prompt: **O/S DIST?**

keystrokes:

[1] [4] [3] [0] [0] [E]

output:

143+00.000
N = 39,297.8695
E = 30,396.6796

prompt: **O/S DIST?**

keystrokes:

output:

144+00.000
N = 39,397.8630
E = 30,395.8093

prompt: **O/S DIST?**

keystrokes:

[1] [4] [4] [4] [6] [.]
[8] [0] [7] [E]

output:

144+46.807
N = 39,444.6697
E = 30,395.9750

Auto-Inverse

This routine is similar to the previous one, in that it calculates the coordinates to the solution station and offsets. In addition, it also calculates the angles and distances for radial stakeout in the field.

Working from one point on a control line and sighting another, all of the points which you calculate can be 'sprayed' directly using an EDM for the distances.



Program memory should contain the program "**SP**" and the same utility programs that are used for the ALIGNMENT & OFFSET routine and the DEFLECTION & CHORD routine. Size the calculator to at least 050.

Because the steps are identical to the ALIGNMENT & OFFSET routine, with the exception of the input for the setup and backsight, we will give the keystroke instructions through the first prompt for station.

Initialize the program by stroking XEQ ALPHA S P ALPHA

1 **COORD-O/S?**

Answer this prompt yes, stroke Y R/S

2 **BRG=?**
Input the entrance tangent bearing R/S

3 **QD=?**
Input the quadrant code for the bearing **toward** the P.I. of the system R/S

4 **TS N↑E**
Input the north coordinate of the TS ENTER↑

Input the east coordinate of the TS R/S

5 **INVERSE?**
Answer this prompt yes Y R/S

6 **INST N↑E**
Input the north coordinate of the setup point ENTER↑

Input the east coordinate of the setup point R/S

7 **BACKSITE?**
Input the north coordinate of the backsight station ENTER↑

Input the east coordinate of the backsight station R/S

8 **PI STATION?**
Input the station of the system P.I. R/S

9 **DELTA?**
Input the system delta. If the curve is to the left, CHS R/S

10 **R?**
Input the radius for the circular curve R/S

11 **L?**
Input the spiral length R/S

Output of the curve data will be next. R/S if no printer is attached. This is followed by the first **STA?** prompt.

Keystroke Examples

As an example of the keystrokes used with this routine, we will follow the steps through the input of the first station, and do one set of the offsets at that station. There is really no difference in the keystrokes between this routine and the previous one, except for the input of the setup and backsight coordinates.

When you answer yes to the prompt, "INVERSE", the prompts for coordinate input are called up. After that input the keystrokes are the same as for the Alignment & Offset routine.

keystrokes:
XEQ ALPHA S P ALPHA

prompt: **COORD-O/S?**

keystrokes:
Y R/S

prompt: **BRG=?**

keystrokes:
2 1 . 5 1 3 R/S

prompt: **QD=?**

keystrokes:
4 R/S

prompt: **TS N↑E**

keystrokes:
3 8 6 3 8 .
3 8 8 ENTER↑
3 0 5 4 9 .
7 9 2 R/S

prompt: **INVERSE?**

keystrokes:
Y R/S

prompt: **INST N↑E**

keystrokes:
3 9 1 0 0
ENTER↑
3 0 6 2 5
R/S

prompt: **BACKSITE?**

keystrokes:
3 8 7 0 0
ENTER↑
3 1 0 0 0
R/S

prompt: **PI STATION?**

keystrokes:
1 4 0 3 6 .
4 4 5 R/S

prompt: **DELTA?**

keystrokes:
2 2 . 0 6 5 5
R/S

prompt: **R?**

keystrokes:

1 4 6 0 R/S

prompt: **L?**

keystrokes:

2 6 5 R/S

output:

```
L = 265.0000
S∠=5°11'59.2"
R = 1,460.0000

PI=140+36.445
CENTRAL ∠ = 22°06'55.0"

TS=136+18.270
SC=138+83.270
```

prompt: **STA?**

keystrokes:

1 3 7 0 0 E

output:

```
137+00.000
N = 38,714.3292
E = 30,519.5814

HD = 399.819
∠RT=58°26'24.3"
```

prompt: **O/S DIST?**

keystrokes:

2 0 CHS R/S

output:

```
O/S = -20.0000
N = 38,707.0434
E = 30,500.9556

HD = 412.070
∠RT=60°40'18.1"
```

prompt: **O/S DIST?**

keystrokes:

2 0 R/S

output:

```
O/S = 20.0000
N = 38,721.6149
E = 30,538.2071

HD = 388.212
∠RT=56°04'16.5"
```

If you would like more practice with this routine, you can refer to the keystroke examples on pages 14 through 16, and continue with the calculations. All of the keystrokes are the same, and the only difference in the output will be the distance and angle to the calculated point.

RADIAL INVERSING

Because the capability for radial inversing is already contained in this program, we have included a routine for inversing from a known setup point to **any** point with known coordinates. This routine may be used by itself for calculating radial stakeout.

Radial inverse

The Radial Inverse routine may be used at any time, if the program "**SP**" and its utility programs are in program memory. Unlike the other routines, the RADIAL INVERSE routine is accessed by using "**GTO**" (shift, RCL) instead of "**XEQ**" to bring the program to the top of program memory. After that, stroke **a** (shift, A) to begin the prompts.

Initialize the program with [ ] **RCL** **ALPHA** **S** **P** **ALPHA** [ ] **A**

1 **INST N↑E**

Input the north coordinate of the instrument point **ENTER↑**

input the east coordinate of the instrument point **R/S**

2 **BACKSITE?**

Input the north coordinate of the backsight point **ENTER↑**

Input the east coordinate of the backsight point **R/S**

3 **N↑E**

Input the north coordinate of the point you wish to set **ENTER↑**

Input the east coordinate of the point you wish to set. **A**

Note that stroking **A** after the input of the east coordinate produces the output. If you do not have a printer attached, stroke **R/S** until the prompt **N↑E** appears.

For the keystroke example on the next page, we will use the information shown to the right.



This keystroke example assumes that the instrument is set up over a point which has the coordinates 100/100, and that we are backsighting a point which has the coordinates 600/80.

The coordinates of the building corners which we want to set are shown in the sketch on page 21. The keystrokes would be as follows:

keystrokes:

[ ] RCL ALPHA S P ALPHA

pause, then

[ ] A

prompt: **INST N↑E**

keystrokes:

1 0 0 ENTER↑

1 0 0 R/S

prompt: **BACKSIGHT?**

keystrokes:

6 0 0 ENTER↑

8 0 R/S

prompt: **N↑E**

keystrokes:

4 2 5 ENTER↑

3 5 0 A

output:

```
425.0000
350.0000

HD = 410.030
∠RT=39°51'33.1"
```

prompt: **N↑E**

keystrokes:

4 0 0 ENTER↑

3 5 0 A

output:

```
400.0000
350.0000

HD = 390.512
∠RT=42°05'46.3"
```

prompt: **N↑E**

keystrokes:

4 2 5 ENTER↑

4 2 0 A

output:

```
425.0000
420.0000

HD = 456.098
∠RT=46°50'47.3"
```

prompt: **N↑E**

keystrokes:

4 0 0 ENTER↑

4 2 0 A

output:

```
400.0000
420.0000

HD = 438.634
∠RT=49°08'17.6"
```

NOTE: The coordinates are not output (as shown) unless the printer is attached.

Tangent Offset

Another common method of layout for spirals is the TANGENT-OFFSET method. This routine calculates the distance along the tangent and the offset from the tangent for any station on the spiral.



Initialize the program by stroking [XEQ] [ALPHA] [S] [P] [ALPHA]

1 **COORD-O/S?**
Answer this prompt no, with [N] [R/S]

2 **TAN O/S?**
Answer this prompt yes, stroke [Y] [R/S]

3 **PI STATION?**
Input the station of the system P.I. [R/S]

4 **DELTA?**
Input the system delta angle. If left, [CHS] before stroking [R/S]

5 **R?**
Input the radius of the circular curve [R/S]

6 **L?**
Input the length of spiral [R/S]

Output will be the spiral system data, as in the previous routines. Continue stroking [R/S], if there is no printer attached, until the "STA?" prompt appears

7 **STA?**

Input the station for which the tangent distance and offset are wanted **E**

Output will be the tangent distance and the tangent offset. Stroke (if no printer) until you again see the prompt "STA?".

8 **STA?**

Repeat step 7 until all of the stations needed in the entrance spiral have been calculated, then go to the exit spiral by stroking **D**

Output will be the CS and ST stations, then the prompt

9 **STA?**

Input the exit spiral stations for solution in the same manner as before, repeating step 7 until all of the required stations have been calculated. The exit spiral can be calculated in either direction, but the tangent distances and offsets are from the ST, sighting toward the P.I.

You should be familiar with the curve input by now, but we'll include it to have a complete example.

keystrokes:

XEQ **ALPHA** **S** **P** **ALPHA**

prompt: **COORD-O/S?**

keystrokes:

N **R/S**

prompt: **TAN O/S?**

keystrokes:

Y **R/S**

prompt: **PI STATION?**

keystrokes:

1 **4** **0** **3** **6** **·** **4** **4** **5** **R/S**

prompt: **DELTA?**

keystrokes:

2 **2** **·** **0** **6** **5** **5** **R/S**

prompt: **R?**

keystrokes:

1 **4** **6** **0** **R/S**

prompt: **L?**

keystrokes:

2 6 5 R/S

output:

L = 265.0000
S∡=5 11 59.2
R = 1,460.0000

PI=140+36.445
CENTRAL ∡ = 22 06 55.0

TS=136+18.270
SC=138+83.270

prompt: **STA?**

keystrokes:

1 3 7 0 0
E

output:

137+00.000
TD = 81.7294
T O/S = 0.2352

prompt: **STA?**

keystrokes:

1 3 8 0 0 E

output:

138+00.000
TD = 181.6969
T O/S = 2.5851

prompt: **STA?**

keystrokes:

1 3 8 8 3 ·
2 7 E

output:

138+83.270
TD = 264.7818
T O/S = 8.0118

With the entrance spiral calculated, move to the exit spiral with the keystroke

D

output:

ST=144+46.807
CS=141+81.807

prompt: **STA?**

keystrokes:

1 4 1 8 1 ·
8 0 7 E

output:

141+81.807
TD = 264.7821
T O/S = -8.0119

prompt: **STA?**

keystrokes:

1 4 2 0 0 E

output:

142+00.000
TD = 246.6544
T O/S = -6.4734

prompt: **STA?**

keystrokes:

1 4 3 0 0 E

output:

143+00.000
TD = 146.7959
T O/S = -1.3629

prompt: **STA?**

keystrokes:

1 4 4 0 0 E

output:

144+00.000
TD = 46.8072
T O/S = -0.0442



THE RADIUS POINTS ARE
"DOWNSTATION"
OF THESE TWO INTERSECTIONS
Radius
Point
CS
ST
SC
Radius
point
THE RADIUS POINTS ARE
"UPSTATION"
OF THESE TWO INTERSECTIONS
TS
Stationing
SPIRAL AND CIRCULAR
CURVE INTERSECTION
Known
Point
KNOWN BEARING
ST
"OUTSIDE"
CS
KNOWN BEARING
SC
"INSIDE"
KNOWN BEARING
SPIRAL CURVE AND
LINE INTERSECTION
KNOWN BEARING
Known
Point
Stationing
TS

Intersections

We'll be the first to tell you that this is not a **fast** program, but if you've ever had to trial-and-error your way to an intersection point along a spiral, you'll be happy that you have a program which will even do spiral intersections.

The program will calculate the spiral station and coordinates for an intersection with a line which comes from a known point at a known bearing, or with a curve whose radius and radius point coordinates are known.

The routine is accessed through the COORD-O/S? prompt, with the same input of the spiral data, to the first "STA?" prompt.

Initialize the program by stroking XEQ ALPHA S P ALPHA

1 **COORD-O/S?**
Answer this prompt yes, stroke Y R/S

2 **BRG=?**
Input the entrance tangent bearing, and stroke R/S

3 **QD=?**
Input the quadrant code for the bearing **toward** the P.I. R/S

4 **TS N↑E**
Input the north coordinate of the TS ENTER↑

Input the east coordinate of the TS R/S

5 **PI STATION?**
Input the main P.I. station R/S

6 **DELTA?**
Input the system delta. If curve left, CHS before stroking R/S

7 **R?**
Input the radius of the system's circular curve R/S

8 **L?**

Input the length of the spiral R/S

Output will be the spiral data and the TS and SC stations.

9a **STA?**

If the intersection occurs in the entrance spiral, enter the INTERSECTION ROUTINE by stroking I

or

9b If the intersection occurs in the exit spiral, stroke D. After the output of the SC and ST stations, at the "STA" prompt, stroke I

10 **POINT N↑E**

Input the north coordinate of the known (or radius) point ENTER↑

Input the east coordinate of the known (or radius) point R/S

11 **APPROX STA?**

Input the approximate station along the spiral where the intersection occurs. This may be scaled. R/S

12a **CURVE?**

If this is a spiral to curve intersection stroke Y R/S

GO TO STEP 13C

or

12b If this is a line to spiral intersection, stroke N R/S

For the line intersection, you will need to know whether the origin is "inside" or "outside" of the curve, as defined in the sketch on page 26.

13L **OUT?**

If the point lies "outside" stroke **Y**, if "inside" stroke **N**

R/S

14L **BRG=?**

Input the bearing of the line

R/S

15L **QD=?**

Input the quadrant code **from the spiral to the point**

R/S

Output will be the distance from the point to the spiral, followed by the spiral station and its coordinates.

The curve intersection requires knowing whether the radius point is "upstation" or "downstation" from the intersection (see sketch on page 26).

13C **UP?**

Stroke **Y** if the point lies "upstation", or **N** if "downstation"

R/S

14C **R?**

Input the radius of the intersecting curve

R/S

Output will be the azimuth **from the radius point** to the spiral, followed by the spiral station and its coordinates.

In both cases, the program takes the approximate station, tries it, and then adjusts the station and tries again.

This process is continued until a matchup is achieved, and the answer is output. The station will be accurate within ±0.012 feet, and the azimuth output will be within ±2.5 seconds (based on comparisons with known data).

More accuracy may be attained (see program listings), but this makes it a **very slow** program, instead of a **not fast** program. If you remember that until recently (computers), we only calculated layout to nearest tenth of a minute and used the approximate $\frac{1}{3}\theta$ as gospel, the accuracy is satisfactory within the limits of the accuracy with which the points can be set.

As an example of a curve to spiral intersection, we have a curve with a radius of 250' intersecting our entrance spiral at a scaled station of 137+54.

The coordinates of the known radius point are N38,573.5972 and E30,663.4123. The radius point is what we call "downstation" from the intersection.

Find the actual station of the intersection, and the azimuth of the circular radius at the point of intersection.



keystrokes:

XEQ ALPHA S P ALPHA

prompt: **COORD-O/S?**

keystrokes:

Y R/S

prompt: **BRG=?**

keystrokes:

2 1 . 5 1 3 R/S

prompt: **QD=?**

keystrokes:

4 R/S

prompt: **TS N↑E**

keystrokes:

3 8 6 3 8 .
3 8 8 ENTER↑
3 0 5 4 9 .
7 9 2 R/S

prompt: **INVERSE?**

keystrokes:

N R/S

prompt: **PI STATION?**

keystrokes:

1 4 0 3 6 .
4 4 5 R/S

prompt: **DELTA?**

keystrokes:

2 2 . 0 6 5 5
R/S

prompt: **R?**

keystrokes:

1 4 6 0 R/S

prompt: **L?**

keystrokes:

2 6 5 R/S

output:

L = 265.0000

SC=138+83.270

Keystroke Examples

prompt: **STA?**

keystrokes:

I

prompt: **POINT N↑E?**

keystrokes:

3 8 5 7 3 .
5 9 7 2 ENTER↑
3 0 6 6 3 .
4 1 2 3 R/S

prompt: **APPROX STA?**

keystrokes:

1 3 7 5 4 R/S

prompt: **CURVE?**

keystrokes:

Y R/S

prompt: **UP?**

keystrokes:

N R/S

prompt: **R?**

keystrokes:

2 5 0 R/S

output:

319 24 35.1

137+52.600
N = 38,763.4428
E = 30,500.7509

The actual azimuth used to design the example was 319°24'35", and the station was 137+52.6. We can use the same point and try a line intersection; use a bearing of S 40°35'25 E, and start by stroking I (more than one intersection may be calculated after input of the spiral data).

prompt: **POINT N↑E?**

keystrokes:

3 8 5 7 3 .
5 9 7 2 ENTER↑
3 0 6 6 3 .
4 1 2 3 R/S

prompt: **APPROX STA?**

keystrokes:

1 3 7 5 4 R/S

prompt: **CURVE?**

keystrokes:

N R/S

prompt: **OUT?**

keystrokes:

N R/S

prompt: **BRG=?**

keystrokes:

4 0 . 3 5 2 5
R/S

prompt: **QD=?**

keystrokes:

2 R/S

output:

250.009

137+52.610
N = 38,763.4515
E = 30,500.7477

Program Listings

These programs, "**SP**" and the utility programs used with it, contain 1658 bytes of programming (834 program steps), and occupy 289 registers of program memory when sized to the minimum 050.

If your version of this booklet contains the programmed cards, angular output will appear as 22°06'55.0". As shown in the program listings (no synthetics) the angular output will be as 22 06 55.0.

As mentioned on page 29, more accuracy may be attained in the intersection routines by changing **FIX 3** to **FIX 4** at program steps 342 and 618, but this is not advised.

```
01♦LBL "SP"
02 CF 02
03 SF 21
04 SF 27
05 CLRG
06 FIX 4
07 CF 01
08 3
09 STO 11
10 -10
11 STO 12
12 -42
13 STO 13
14 216
15 STO 14
16 1320
17 STO 15
18 -9360
19 STO 16
20 -75600
21 STO 17
22 685440
23 STO 18
24 6894720
25 STO 19
26 CF 22
27 CF 03
28 CF 14
29 CF 15
30 CF 05
31 CF 06
32 CF 07
33 CF 08
34 CF 09
35 XEQ 21
36 FS? 05
37 XEQ 22
38 FS? 03
39 XEQ 17
40♦LBL 01
41 "PI STATION?"
42 PROMPT
43 STO 06
44 "DELTA?"
45 PROMPT
46 X<0?
47 XEQ 30
48 HR
49 2
50 /
51 STO 05
52 CF 01
53 "R?"
54 PROMPT
55 CF 04
56 STO 20
57 "L=?"
58 PROMPT
59 STO 40
60 ENTER↑
61 ST+ 02
62 RCL 20
63 /
64 2
65 /
66 STO 00
67 R-D
68 STO 41
69 HMS
70 "L = "
71 ARCL 40
72 AVIEW
73 "S∡="
74 XEQ "DMS"
75 "R = "
76 ARCL 20
77 AVIEW
78 CLD
79 ADV
80 FS? 01
81 GTO 15
82 0
83 STO 01
84 RCL 40
85 STO 02
86 XEQ 09
87 RCL 41
88 RCL 20
89 P-R
90 RDN
91 -
92 RDN
93 +
94 RCL 05
95 TAN
96 *
97 R↑
98 +
99 STO 42
100 RCL 06
101 X<>Y
102 -
103 STO 01
104 ST+ 02
105 RCL 05
106 RCL 41
107 -
108 2
109 *
110 STO 21
111 D-R
112 RCL 20
113 *
114 RCL 40
115 2
116 *
117 +
118 +
119 STO 07
120 RCL 06
121 RCL 01
122 0
123 RDN
124 CLA
125 "PI="
126 RDN
127 SF 10
128 XEQ "STA"
129 CLA
130 FIX 4
```

```
131 "CENTRAL ∡ = "
132 RCL 05
133 2
134 *
135 HMS
136 XEQ "DMS"
137 CLD
138♦LBL 15
139 XEQ 09
140 X<>Y
141 RCL 41
142 TAN
143 /
144 -
145 FS? 02
146 CHS
147 STO 42
148 0
149 RCL 01
150 RCL 40
151 +
152 STO 03
153 RCL 01
154 RCL 42
155 +
156 RCL 01
157 ADV
158 "TS="
159 FS? 02
160 "ST="
161 RCL 01
162 SF 10
163 XEQ "STA"
164 CLA
165 "SC="
166 FS? 02
167 "CS="
168 RCL 03
169 SF 10
170 XEQ "STA"
171 "STA?"
172 PROMPT
173♦LBL D
174 SF 08
175 CLA
176 FS? 05
177 XEQ 22
178♦LBL 18
179 RCL 07
180 RCL 01
181 STO 07
182 X<>Y
183 STO 01
184 RCL 40
185 CHS
186 STO 40
187 +
188 STO 02
189 SF 02
190 GTO 15
191♦LBL E
192 STO 02
193 CF 01
194 FC? 11
195 ADV
196♦LBL 19
197 SF 04
198 RCL 02
199 -
200 FS?C 22
201 GTO 00
202 RCL 04
203 CF 02
204 X<0?
205 SF 02
206 ABS
207 RCL 03
208 RCL 02
209 -
210 ABS
211 X<=Y?
212 GTO 00
213 X<>Y
214 GTO 07
215♦LBL 00
216 SF 00
217♦LBL 07
218 FS?C 02
219 CHS
220 ST+ 02
221 RCL 02
222 RCL 01
223 -
224 RCL 40
225 /
226 X↑2
227 RCL 41
228 D-R
229 *
230 STO 00
231 XEQ 09
232 FS? 01
233 GTO 04
234 R-P
235 X<>Y
236 HMS
237♦LBL 04
238 STO 30
239 X<>Y
240 STO 31
241 0
242 STO 32
243 RCL 02
244 FC? 11
245 XEQ "STA"
246 FIX 3
247 "CD = "
248 ARCL 31
249 FS? 14
250 AVIEW
251 RCL 30
252 "DEFLECTION ∡="
253 FS? 14
254 XEQ "DMS"
255 RCL 30
256 CHS
257 RCL 00
258 R-D
259 HMS
260 HMS+
261 90
262 HMS+
263 "RADIAL ∡ ="
264 FS? 14
265 XEQ "DMS"
266 HR
267 STO 38
268 FC? 03
269 CLD
270 FS? 05
271 GTO 23
272 FS? 03
273 GTO 23
274 "STA?"
275 FS? 14
276 PROMPT
277 FS?C 00
278 GTO 19
279♦LBL 09
280 CF 22
281 9
282 STO 43
283 0
284 STO 08
285 STO 09
286♦LBL 02
287 RCL 00
288 RCL 43
289 Y↑X
290 10
291 ST+ 43
292 CLX
293 RDN
294 RCL IND 43
295 /
296 10
297 ST- 43
298 CLX
299 RDN
300 RCL 08
301 RCL 09
302 STO 08
303 RDN
304 +
305 STO 09
306 DSE 43
```

```
307 GTO 02
308 RCL 09
309 RCL 02
310 RCL 01
311 -
312 ABS
313 *
314 LASTX
315 RCL 08
316 1
317 +
318 *
319 RTN
320♦LBL I
321 SF 11
322 "POINT N↑E?"
323 PROMPT
324 STO 49
325 RDN
326 STO 48
327 "APPROX STA?"
328 PROMPT
329 STO 02
330 "CURVE"
331 XEQ 25
332 FS? 10
333 GTO 10
334 "OUT"
335 XEQ 25
336 FS? 10
337 CF 26
338 SF 10
339 XEQ "AZ"
340 CF 10
341♦LBL 12
342 FIX 3
343 RND
344 STO 24
345 RCL 02
346 XEQ E
347 RTN
348♦LBL 10
349 SF 12
350 "UP"
351 XEQ 25
352 FS? 10
353 CF 26
354 CF 10
355 "R?"
356 PROMPT
357 XEQ 12
358 RTN
359♦LBL 21
360 "COORD-O/S"
361 XEQ 25
362 FS? 10
363 SF 05
364 CF 10
365 FS? 05
366 XEQ 22
367 FS? 05
368 GTO 16
369 "TAN O/S"
370 XEQ 25
371 FS? 10
372 SF 03
373 CF 10
374 FS? 03
375 GTO 17
376 SF 14
377 RTN
378♦LBL 16
379 "INVERSE"
380 XEQ 25
381 FS? 10
382 SF 06
383 CF 10
384 FS? 06
385 XEQ 20
386 GTO 01
387 RTN
388♦LBL 25
389 "⊢?"
390 AON
391 PROMPT
392 AOFF
393 ASTO X
394 "Y"
395 ASTO Y
396 CLA
397 CF 10
398 X=Y?
399 SF 10
400 RTN
401♦LBL 17
402 0
403 STO 44
404 STO 33
405 STO 39
406 GTO 01
407 RTN
408♦LBL a
409 SF 15
410♦LBL 20
411 "INST N↑E"
412 PROMPT
413 STO 49
414 RDN
415 STO 48
416 "BACKSITE?"
417 PROMPT
418 RCL 49
419 -
420 X<>Y
421 RCL 48
422 -
423 R-P
424 CLX
425 X<>Y
426 X<0?
427 360
428 +
429 HMS
430 STO 45
431 "N↑ E"
432 FS? 15
433 PROMPT
434 GTO 01
435 RTN
436♦LBL A
437 ADV
438 CLA
439 ARCL Y
440 FS? 55
441 AVIEW
442 CLA
443 ARCL X
444 FS? 55
445 AVIEW
446 XEQ 03
447 "N↑ E?"
448 PROMPT
449 RTN
450♦LBL 03
451 ADV
452 RCL 49
453 -
454 X<>Y
455 RCL 48
456 -
457 R-P
458 FIX 3
459 "HD = "
460 ARCL X
461 AVIEW
462 CLX
463 X<>Y
464 X<0?
465 360
466 +
467 STO 46
468 ENTER↑
469 ENTER↑
470 90
471 /
472 1
473 +
474 INT
475 STO 47
476 2
477 /
478 INT
479 180
480 *
481 -
482 ABS
```

```
483 HMS
484 FIX 4
485 RCL 47
486 RCL 46
487 RCL 45
488 HR
489 -
490 ENTER↑
491 CLX
492 X<>Y
493 X<0?
494 360
495 +
496 HMS
497 "∡RT="
498 XEQ "DMS"
499 RTN
500♦LBL 30
501 SF 07
502 CHS
503 RTN
504♦LBL 22
505 SF 10
506 XEQ "AZ"
507 CF 10
508 STO 39
509 "TS N↑E"
510 FS? 08
511 "ST N↑E"
512 PROMPT
513 STO 33
514 RDN
515 STO 44
516 RTN
517♦LBL 23
518 RCL 39
519 RCL 30
520 HR
521 FS? 07
522 CHS
523 FS? 08
524 CHS
525 +
526 STO 37
527 RCL 31
528 P-R
529 RCL 44
530 +
531 STO 35
532 X<>Y
533 RCL 33
534 +
535 STO 36
536 FIX 4
537 "N = "
538 FS? 03
539 "TD = "
540 ARCL 35
541 FC? 11
542 AVIEW
543 "E = "
544 FS? 03
545 "T O/S = "
546 ARCL 36
547 FC? 11
548 AVIEW
549 FS? 06
550 XEQ 03
551 RCL 37
552 RCL 38
553 FS? 07
554 XEQ 31
555 FS? 08
556 CHS
557 +
558 STO 32
559 FS? 11
560 XEQ 05
561♦LBL 24
562 ADV
563 RCL 32
564 "STA?"
565 FS? 03
566 PROMPT
567 "O/S DIST?"
568 FC? 11
569 PROMPT
570 "O/S = "
571 FS? 09
572 "RADIUS POINT "
573 FC? 09
574 ARCL X
575 FS? 55
576 AVIEW
577 P-R
578 RCL 35
579 +
580 FS? 09
581 STO 35
582 X<>Y
583 RCL 36
584 +
585 FS? 09
586 STO 36
587 "N = "
588 ARCL Y
589 AVIEW
590 "E = "
591 ARCL X
592 AVIEW
593 FS? 06
594 XEQ 28
595 FS? 09
596 GTO 32
597 GTO 24
598 RTN
599♦LBL E
600 GTO E
601 RTN
602♦LBL 05
603 RCL 36
604 RCL 49
605 -
606 RCL 35
607 RCL 48
608 -
609 R-P
610 FS? 12
611 X<>Y
612 STO 25
613 X<>Y
614 FC? 12
615 180
616 FC? 12
617 +
618 FIX 3
619 RND
620 RCL 24
621 X=Y?
622 GTO 06
623 FS? 12
624 XEQ 13
625 -
626 D-R
627 RCL 20
628 RCL 31
629 X↑2
630 RCL 40
631 X↑2
632 /
633 *
634 *
635 .8
636 *
637 FS? 26
638 CHS
639 RCL 02
640 +
641 XEQ E
642 RTN
643♦LBL 13
644 -
645 ENTER↑
646 ATAN
647 COS
648 /
649 FS? 26
650 CHS
651 RCL 02
652 +
653 XEQ E
654 RTN
655♦LBL 06
656 CF 11
657 FIX 3
658 RCL 25
```

```
659 CLA
660 ARCL X
661 FC? 12
662 AVIEW
663 CLA
664 FS? 12
665 XEQ 14
666 SF 26
667 RCL 02
668 XEQ E
669 RTN
670♦LBL 14
671 CF 12
672 0
673 X<>Y
674 X<Y?
675 360
676 +
677 HMS
678 XEQ "DMS"
679 CLA
680 RTN
681♦LBL J
682 SF 09
683 "CIRCULAR:"
684 FS? 55
685 AVIEW
686 GTO 24
687♦LBL 31
688 CHS
689 180
690 +
691 RTN
692♦LBL 32
693 "STA?"
694 PROMPT
695 ADV
696 FS? 55
697 XEQ "STA"
698 FIX 4
699 RCL 03
700 -
701 RCL 32
702 FC? 07
703 180
704 FC? 07
705 -
706 X<>Y
707 180
708 *
709 PI
710 /
711 RCL 20
712 /
713 FS? 07
714 CHS
715 +
716 RCL 20
717 "O/S = "
718 PROMPT
719 ARCL X
720 X≠0?
721 AVIEW
722 FC? 07
723 CHS
724 +
725 P-R
726 RCL 35
727 +
728 X<>Y
729 RCL 36
730 +
731 "N = "
732 ARCL Y
733 AVIEW
734 "E = "
735 ARCL X
736 AVIEW
737 FS? 06
738 XEQ 03
739 CF 09
740 GTO 32
741 RTN
742♦LBL 28
743 FC? 09
744 XEQ 03
745 RTN
746♦LBL "DMS"
747 ENTER↑
748 INT
749 CF 29
750 FIX 0
751 ARCL X
752 -
753 1 E2
754 *
755 ABS
756 STO 23
757 INT
758 "├ "
759 XEQ 08
760 ARCL X
761 RCL 23
762 FRC
763 1 E2
764 *
765 FIX 1
766 "├ "
767 XEQ 08
768 ARCL X
769 AVIEW
770 FIX 4
771 SF 29
772 RTN
773♦LBL 08
774 10
775 X<>Y
776 X<Y?
777 "├0"
778 RTN
779♦LBL "STA"
780 CF 29
781 FIX 0
782 FC? 10
783 CLA
784 STO Y
785 1 E2
786 /
787 ENTER↑
788 INT
789 ARCL X
790 -
791 "├+"
792 FIX 3
793 1 E2
794 *
795 RND
796 10
797 X>Y?
798 "├0"
799 ARCL Y
800 R↑
801 R↑
802 AVIEW
803 SF 29
804 FIX 4
805 CF 10
806 RTN
807♦LBL "AZ"
808 "BRG=?"
809 PROMPT
810 "QD=?"
811 PROMPT
812 X<>Y
813 HR
814 X<>Y
815 ENTER↑
816 ENTER↑
817 2
818 /
819 INT
820 PI
821 R-D
822 *
823 X<>Y
824 LASTX
825 *
826 COS
827 R↑
828 *
829 -
830 FS? 10
831 RTN
832 HMS
833 RTN
834 END
```

HP41-CV/CX
SOLUTIONS BOOKLETS

VERTICAL ALIGNMENT
Calculates CONTINUOUS vertical alignment without changing back and forth between Grade and Curve routines. Asymmetrical, Compound and Reverse Curves.

booklet #525 . . . . . **$6.95**
with programmed cards **$12.95**

SPIRAL CURVE SOLUTIONS
The only COMPLETE spiral curve program for use with hand-helds.

Booklet #529 . . . . . **$6.95**
with programmed cards **$12.95**

EDM SLOPE STAKING
Sets slope stakes and reference points from remote instrument and backsight locations directly.

Booklet #533 . . . . . **$6.95**
with programmed cards **$12.95**

TRIANGLE SOLUTIONS
The most complete triangle solutions program available. Solves with any of the following knowns: ASA SAA SAS SSA SSS Area-SS Area-AA Area-SA

Booklet #535 . . . . . **$6.95**
with programmed cards **$12.95**

POINT STORAGE CONVERSIONS
for the
HP41-CV/CX Surveying Pac

Traverse, Inverse & Sideshots
w/automatic compass correction.

Booklet #521 . . . . . **$6.95**
with programmed cards **$12.95**

INTERSECTION SOLUTIONS
(Auto-Inverse routine added) Bearing-Bearing, Bearing-Dist, Dist-Dist and Offset to a Line.

Booklet #523 . . . . . **$6.95**
with programmed cards **$12.95**

Coordinate Transformation
Rotate or re-scale. Automatically outputs the new rotated bearings and distances between points.

Booklet #527 . . . . . **$6.95**
with programmed cards **$12.95**

PREDETERMINED AREAS
Calculate required areas from your pre-stored coordinate pairs.

Booklet #531 . . . . . **$6.95**
with programmed cards **$12.95**

Software by D'Zign
P. O. BOX 1370 • PACIFICA, CA 94044
(415) 355-8942

CA residents add 6.5% sales tax

COGO 41

This is the ROM you've been needing. Complete traverse, inverse . . . .everything the others have and more! Coordinate storage and retrieval option by **point number**. No need to pre-divide to mean angles before input in the field.

The closure routine includes **automatic angle check and adjustment**. The compass and transit corrections are automatic, with **auto-inverse** between the adjusted coordinates, all at the touch of a button. Does not require X-memory to work.

Contains keyboard functions for all of the common conversions; Feet to Meters, Feet to Foot-Inch-Fraction, Azimuth to Bearing, (option of North or South Azimuth), add or subtract feet-inch-fractions directly with one shifted keystroke! And, of course, inverse functions of all of these.

available early 1988

$6.95

41 888

ISBN 0-944889-00-X